



# ASMForth II

For the TI-99 and Camel99 Forth

02.Mar.2023

Version 0.1

---

Brian Fox  
Kilworth Ontario  
[brian.fox@brianfox.ca](mailto:brian.fox@brianfox.ca)



## Overview

ASMFORTH II is an experimental Machine Forth for the TI-99. It uses Forth-like syntax but is really an Assembler. Many of the words are aliases of the Forth assembler in Camel99 Forth. Although the syntax looks like Forth the significant difference is that registers are referenced explicitly for maximum performance. The Forth data stack and return stack are still available to the programmer but also must be referenced explicitly in your ASMForth code.

## Justification

The author has spent a considerable amount of time adapting Chuck Moore's Machine Forth concept to the TMS9900 and the results are good but not great. The hypothesis is that the lack-lustre performance of Machine Forth on the 9900 is due to the hardware mis-match between the 9900 and the F21 Forth CPU. Machine Forth is actually the Assembly language for Chuck's F21. The conclusion is that like the original machine Forth, any machine Forth must leverage underlying hardware to be efficient.

## Motivation

ASMForth was originally a curiosity project to see if a 9900 Assembler could be written that used Forth type syntax. It worked and was put on the shelf.

ASMForth II was created when comparing the performance of the Byte Magazine Sieve of Eratosthenes benchmark using conventional Forth, a GCC compiled 'C' version and a TMS 9900 Assembly language version. Here are the timings:

- Camel99 Forth (Indirect threaded) 120 seconds
- C compiled with GCC -O2 optimizing 14 seconds
- Hand coded Assembly language 10 seconds

A version of the sieve using ASMForth II for the inner loop and Forth for the outer loop did the benchmark in 10 seconds.



## ASMForth Fundamentals

AsmForth II gives the programmer access to the full instruction set of the TMS9900. In many cases there is a one-to-one relationship between a Forth word and an instruction. In many other cases the instruction is completely unique to the 9900 CPU and is provided as a TI mnemonic that operates with reverse-polish-notation syntax. (RPN) TI mnemonics and ASMForth words can be interleaved in a program freely.

### Registers are Required

In Forth the data stack is used to provide a simple way to move parameters to and from different blocks of code. The 9900 CPU has no native stack and so the 'PUSH' operation requires two instructions. The POP operation is one instruction. Both however require slower indirect addressing modes and so are expensive on the 9900.

Registers are where performance is gained on this CPU so ASMForth programming uses them. This makes it a bit like using ANS Forth local variables except that the variables are CPU registers. In many cases the simplicity of using registers is easier to understand than "stackrobatics" using SWAP DUP OVER and the rest.

Where the data stack and return stack are useful is for passing parameters from Forth to ASMForth and getting results back to the Forth system. The return stack also lets us create nestable subroutines and loops with ease.

### Registers Allocations

The data stack can be used to save/restore registers for nesting calls. The return stack can also be used to save registers BUT they must be restored before exiting a routine.

The following six registers are free for ASMForth programs to use. R4 doubles as the cache for the top of the data stack. Simply PUSH it to free it up. You can also PUSH any other registers that you want to preserve and POP them back later.

- R0 free to use
- R1 free to use
- R2 free to use
- R3 free to use
- R4 TOS Accumulator
- R5 free to use (used by UM\* with operations on the DATA stack)



These registers have special purposes in the Forth virtual machine. R8 is only a temp register in the interpreter so you can use it as well. Registers marked with '\*\*' are critical if you plan to return to the Forth environment.

SP data stack pointer \*\*  
 RP return stack pointer  
 R8 free to use (Forth working register)  
 R9 Forth interpreter pointer \*\*  
 R10 holds NEXT address (Forth interpreter) \*\*  
 R11 subroutine linkage \*\* use it but be careful  
 R12 CRU I/O \*\* use it but be careful

These registers are completely free if you don't use multitasking with MTASK99 .

R13 Multi-tasking -or- temp  
 R14 Multi-tasking -or- temp  
 R15 Multi-tasking -or- temp

## RPUSH and RPOP are Your Friends

If you really need to use Forth critical Registers you can code in the INTEL CPU style by using RPUSH and RPOP. These instructions let you save registers on the return stack. This let's you use the data stack as normal. At the end of the code you restore the registers with RPOP in reverse order to how they were RPUSHed. It costs cycles but it's there if needed.

Example:

```
: MYWORD
  \ Save these specific registers
    R10 RPUSH
    R11 RPUSH
    R12 RPUSH

    ( Your code goes here )

  \ Restore registers in reverse order
    R12 RPOP
    R11 RPOP
    R10 RPOP
;
```



## High-Level Additions

Since ASMForth is a compiler, albeit a simple one, it provides the programmer with a few goodies that would need to be hand coded in conventional Assembly Language.

## Structured Looping and Branching

Conventional Forth structures are used for loops

- BEGIN UNTIL**                      Jump to BEGIN until condition is true.
- BEGIN WHILE REPEAT**    Perform code after WHILE while condition is true. Jump to BEGIN
- BEGIN AGAIN**                    Jump unconditionally to BEGIN

### FOR NEXT

Execute code between FOR and NEXT for the number of iterations given in the TOS register.

Used in Moore's machine Forth it is a simple down-counting loop. Count is maintained on the return stack so FOR NEXT can be nested. If you want to access the loop counter you can "fetch" the return stack pointer register, RP. ( Example: RP @ TOS ! )

### IF ELSE THEN

These are very similar to standard Forth with one important difference. In Forth the top of the data stack is used to determine true or false status. In ASMForth the CPU status register is used to determine how the jumps proceed. The 'IF' word is actually a "compiler" that compiles the correct jump instruction depending on the following comparison tokens.

### Comparison Tokens

Signed comparisons: = <> < >

Unsigned Comparisons: U> U< U<= U>=

Each machine instruction will cause the status register in the CPU to record the results of that instruction. For maximum performance ASMForth uses that native hardware to jump.

In some cases where you need an explicit comparison the CPU provides three comparison operations. The ASMForth names are below;

- **CMP**                      compare two 16 bit arguments
- **CMPB**                    compare 8 bit arguments



- `#CMP`      compare a 16 bit argument to a literal number

All these words are limited to the constraints of the jump instructions of the 9900 CPU.

+128/-127 cells are the maximum distances they can branch to. This is seldom a problem.

**See code examples at the end of the document.**

## Branching Farther

If you require the branch instruction (B) you can create a label and branch to it.

```
CREATE MYLABEL
```

```
    <code here>
```

```
MYLABEL @@ B
```



## Constants

Forth variables and constants are used in ASMForth with only a few things to remember. Constants can be used for instructions that require a LITERAL number listed below:

1. #!
2. #AND
3. #OR
4. #CMP

Load a constant into a register with #! ("number store")

```
X R3 #!
```

Mask a register with constant

```
HEX  
007F CONSTANT MASK
```

```
TOS MASK #AND
```

Compare a register to a constant

```
R2 MASK #CMP  
= IF <true code> ELSE <>false code> THEN
```

Push a CONSTANT onto the DATA stack with the # macro

```
1234 CONSTANT X  
5678 CONSTANT Y
```

```
X # Y #      ( - x y ) \ X and Y are on the data stack  
NOS^ TOS +    \ add them together with + and pop NOS
```



## Variables

Variables in Forth return their address so that fits nicely with an Assembly Language.

We can “fetch” the contents of a Forth variable with '@@'. This invokes the memory features of the 9900 CPU.

The following are legal statements in ASMForth.

HOST

VARIABLE Q

VARIABLE T

CODE TESTMEM

TOS Q @@ ! \ store TOS into memory at Q location

T @@ TOS ! \ fetch data from memory location T into TOS

Q @@ OFF \ reset the memory at location Q to zero

T @@ ON \ set the memory at location T to -1 (Forth true)

;CODE





## “Colon” Definitions (nestable sub-routines)

Forth uses the colon/semi-colon structure to define new words and ASMForth borrows that syntax to create subroutines. These subroutines can be called from a CODE word by simply invoking their name just like normal Forth. The important difference is that a subroutine can call another subroutine because they use the Forth return stack.

Consider the following ASMForth code:

(This is NOT an example of how to efficiently add 2+2.) :-)

```

ASMFORTH           \ change the namespace to use ASMFORTH keywords

\ sub-routines must be declared before they can be used
: SUB1    NOS^ TOS + ; \ add NOS to TOS and POP NOS (remove it)
: SUB2    SUB1 ;      \ call sub1
: SUB3    SUB2 ;      \ call sub2

\ CODE words give Forth an interface to ASMForth
CODE MYADDITION
    2 #      \ push 2 on the DATA stack
    2 #      \ push another 2 onto DATA stack
    SUB3     \ call SUB3 (compiles: BL @SUB3)
;CODE       \ TOS register holds 4  (2+2)

```

## How it works

Behind the curtain ASMForth ‘:’ compiles code to save the CPU linkage register R11 on the return stack. The ‘;’ pops R11 from the Return stack and does a B @R11 instruction.



## Tail-Call Optimization

Something that Charles Moore added to machine Forth was tail-call Optimization.

This can be used if a subroutine calls another subroutine **as the last function** in the code. Tail-call optimization replaces the branch and link instruction with a simple branch and enters that subroutine without pushing R11 onto the return stack. This speeds up the call greatly and the return stack does not grow larger. Less return stack usage can be very important if you write a recursive subroutine in ASMForth.

### Invoking Tail-Call Optimization

In the previous example we could have code the subroutines this way:

```
: SUB1    NOS^ TOS + ; \ add NOS to TOS and POP NOS (remove it)
: SUB2    SUB1 -;      \ call sub1
: SUB3    SUB2 -;      \ call sub2
```

Notice the use of -; at the end of SUB2 and SUB3. This is the command that invokes tail-call optimization. We cannot tail-call optimize SUB1 because it ends with an instruction not a subroutine call. You have been warned. 😊

## How much difference does Tail-Call Optimization Make?

The difference between using semicolon and minus-semicolon is about 30% speed improvement on the call/return time.

See: demo/nesting.fth for a timing on a one million nest/unnest benchmark



## Using the DATA Stack

You are free to use both of Forth's stacks but it is less automatic than standard Forth.

You must manage literal numbers with the # word. TOS is actually R4 so must keep that in mind as well. This is done typically by using the NOS^ word pops the data stack from memory.

The advantage of embracing the native machine is that we can use operations on registers directly which is more efficient on the 9900 CPU.

The advantage of having the stacks is nested calls and passing parameters to subroutines does not take up valuable registers.

Addition Example:

```
45 # 7 #      \ push 2 numbers onto the DATA stack. NOS=45 TOS=7
NOS^ TOS +    \ Add NOS to TOS. NOS^ pops the stack automatically
              \ TOS=52
```

This compiles the following TI Assembler code

```
DECT SP
MOV  R4,*SP
LI   R4,45
DECT SP
MOV  R4,*SP
LI   R4,7
A    *SP+,R4
```

## Subroutine Parameter Passing in/out

Compute two numbers squared.

ASMFORTH

```
: SQUARE ( n - n^2' )
    DUP          \ push TOS onto memory stack
    NOS^ TOS *    \ Add NOS to TOS and pop NOS
;
```

```
CODE PROGRAM      \ this is our entry to ASMForth from HOST Forth
    5 # SQUARE    \ we can call SQUARE without using any registers except TOS
;CODE
```

HOST ( Switch back to Forth dictionary)



# CODE EXAMPLES

## ASMFORTH LOOPS

\ Type COLD to restart ASMFORTH II

HEX

```
CODE DOUNTIL \ smallest loop
    FFFF # \ DUP R4 and put a number into R4
    BEGIN
        TOS 1- \ decrement a register
    = UNTIL \ loop until TOS hits zero
    DROP \ clean the stack
;CODE
```

HEX

```
CODE DOWHILE
    FFFF #
    BEGIN
        TOS 1-
    <> WHILE
        R5 2+
    REPEAT
    DROP
;CODE
```

\ for/next gives you a nestable loop structure  
 \ put the number of iterations into the TOS register (or use #)  
 \ It will run unto loop index <0 (uses the JNC instruction)  
 \ It does not consume a register while running because  
 \ the counter is on the return stack.

```
CODE FORNEXT
    FFFF # FOR
        RP @ R0 ! \ fetch loop index store it in R0
    NEXT
;CODE
```

```
CODE FOREVER \ reset the machine to get out
    BEGIN
        TOS 1+
    AGAIN
;CODE
```



## Accessing Memory

```

COLD                \ restart the compiler

VARIABLE X          \ Forth variables are used seamlessly
VARIABLE Y

CODE FETCHVAR       \ get a variable into register
    Y @@ R0 !      ( @@ specifies "symbolic" addressing to the CPU)
;CODE

CODE REGISTER2VAR
    R1 X @@ !
;CODE

CODE MEM2MEM        \ move memory to memory in one instruction
    X @@ Y @@ !
;CODE

\ Arrays in memory

CREATE Q    400 CELLS ALLOT \ Q returns base address
                                \ 400 CELLS = 800 bytes allocated

CODE FETCH-ARRAY
    6 R1 #!           \ store 6 in R1. R1 is our "index register"
    Q (R1) R4 !       \ fetch contents of Q+R1 into R4
;CODE

```



## Byte Magazine Sieve of Eratosthenes

Changes were made to the original code for ANS/ISO Forth. The leading zero was removed in the VARIABLE statement. We removed the text I/O from the computation loop and put it in the main loop. This fits better with ASMForth which does not have a console I/O library and also suits the author's sense of Forth coding style.

```

DECIMAL
8190 CONSTANT SIZE
VARIABLE FLAGS    SIZE ALLOT  0 FLAGS !

: DO-PRIME
  FLAGS SIZE  1 FILL  ( set array )
  0              ( counter )
  SIZE 0
  DO FLAGS I + C@
    IF I DUP + 3 + DUP I +
      BEGIN
        DUP SIZE <
        WHILE
          0 OVER FLAGS + C!
          OVER +
        REPEAT
        DROP DROP
        1+
      THEN
    LOOP
  ;

: PRIMES ( -- )
  PAGE ." 10 Iterations"
  10 0 DO DO-PRIME CR SPACE . ." Primes" LOOP
  CR ." Done!"
  ;

```



## ASMForth II Sieve Program

This code is a translation of the Assembly language program that follows this code.

HOST

DECIMAL

8190 CONSTANT SIZE

HEX 2000 CONSTANT FLAGS \ array in Low RAM

ASMFORTH

: FILLW ( addr size char --)

R0 POP \ size

R1 POP \ base of array

BEGIN

TOS R1 @+ ! \ write ones to FLAGS

R0 2-

NC UNTIL

DROP ;

HEX

CODE DO-PRIME ( -- n) \ CODE words can be called from Forth

FLAGS # SIZE # 0101 # FILLW

\ inits

R0 OFF \ clear loop index

R3 OFF \ 0 constant

FLAGS R5 #! \ array base address

0 # \ counter on top of Forth stack

SIZE # FOR \ ASMForth II has Chuck's FOR NEXT loop

R5 @+ R3 CMPB \ FLAGS C@+ byte-compared to R3 (ie: 0)

<> IF \ not equal to zero ?

R0 R1 ! \ I -> R1

R1 2\* R1 3 #+ \ R1 3+

R0 R2 ! \ I -> R2 ( R2 is K index)

R1 R2 + \ PRIME K +!

BEGIN

R2 SIZE #CMP \ K SIZE compare

< WHILE

R3 FLAGS (R2) C! \ reset byte, FLAGS(R2)

R1 R2 + \ PRIME K +!

REPEAT

TOS 1+ \ increment no. of primes

THEN

( continues on next page )





```
      R0 1+                \ bump index register
    NEXT
;CODE

\ 10 iteration loop is written in interpreted Forth

: PRIMES ( -- )
  PAGE ." 10 Iterations"
  10 0 DO DO-PRIME CR SPACE . ." Primes" LOOP
  CR ." Done!"
;
```



## Assembly Language Version

For comparison, here is an assembler version written by "Reciprocating Bill" on Atariage.com.

It is clearly the work of an experienced programmer. It includes code to write text and numbers to the TI-99 screen.

```
* TMS9900 Assembly Language Version
* SIEVE OF ERATOSTHENES -----
* WSM 4/2022
* TMS9900 assembly adapted from BYTE magazine 9/81 and 1/83 issues
* 10 iterations 6.4 seconds on 16-bit console
* ~10 seconds on stock console

      DEF    START
      REF    VMBW

M      DATA >0000          for PRINT routine
M2     BSS    6
M3     DATA 10000
      DATA 1000
      DATA 100
      DATA 10
ROW    DATA >0004

MSG1   TEXT ' Sieve of Eratosthenes '
MSG2   TEXT '      10 iterations '
MSG3   TEXT '      done! '

SIZE   EQU   8190
FLAGS  BSS   8191
PRWS   BSS   >20

* main *
      EVEN
START  LWPI >8300          workspace in PAD for stock console
      LI    R5,>0101      to initialize array
      LI    R8,SIZE       index
      LI    R12,10        # iterations
      LI    R15,11        constant (for calculating row of display)

      LI    R0,36         display title
      LI    R1,MSG1
      LI    R2,24
      BLWP  @VMBW
      AI    R0,32
      LI    R1,MSG2
      LI    R2,20
      BLWP  @VMBW
```

AGAIN	CLR	R7	0 constant
	CLR	R10	counts primes
	LI	R6,FLAGS	base of FLAGS array
	LI	R0,SIZE	index FLAGS array
LOOP	MOV	R5,*R6+	write ones to FLAGS
	DECT	R0	
	JNE	LOOP	
FOR	LI	R6,FLAGS	reset index
	CB	*R6+,R7	FLAGS(I)=0?
	JEQ	SIEVE3	yes, skip strikeouts loop
	MOV	R0,R1	PRIME=I
	SLA	R1,1	*2
	AI	R1,3	+3
	MOV	R0,R2	K=I
	A	R1,R2	add PRIME to K
SIEVE1	C	R2,R8	K>SIZE?
	JGT	SIEVE2	yes, goto SIEVE2
	MOVB	R7,@FLAGS(R2)	no,reset @FLAGS(R2)
	A	R1,R2	K=K+PRIME
	JMP	SIEVE1	
SIEVE2	INC	R10	increment count of primes
SIEVE3	INC	R0	next I
	C	R0,R8	I>SIZE?
	JLT	FOR	no, next
	MOV	R10,@M	print #primes found
	BL	@PRINT	
	DEC	R12	
	JNE	AGAIN	rinse and repeat
	MOV	@ROW,R15	print "Done!"
	INCT	R15	
	SLA	R15,5	
	MOV	R15,R0	
	LI	R1,MSG3	
	LI	R2,12	
	BLWP	@VMBW	
OUT	LIMI	2	wait for quit
	JMP	OUT	

(continues on next page)

\* Print binary value as decimal (from Morley, p.149)  
 \* value to print @M

```

PRINT  LWPI PRWS
        INC  @ROW                increment row counter
        LI   R0,4
        LI   R1,M3
        LI   R2,M2                R2 points to address containing result
        MOV  @M,R4                mov integer to be converted to R4
J1      CLR  R3                    see Morley for comments
        DIV  *R1+,R3
        SLA  R3,8
        AI   R3,>3000
        MOVB R3,*R2+
        DEC  R0
        JNE  J1
        SLA  R4,8
        AI   R4,>3000
        MOVB R4,*R2+
        LI   R1,>2000
        MOVB R1,*R2
        LI   R0,>3000
        LI   R3,M2
J2      CB   *R3,R0
        JNE  J3
        MOVB R1,*R3+
        JMP  J2
J3      CB   *R3,R1
        JEQ  J4
        JMP  J5
J4      DEC  R3
        MOVB R0,*R3
J5      MOV  @ROW,R15              use row counter
        SLA  R15,5                row *32
        MOV  R15,R0
        INCT R0                    write to this address in VDP
        LI   R1,M2
        LI   R2,6
        BLWP @VMBW
        LWPI >8300
        RT

        END

```